Fractional-Order Modelling in Modelica

Alexander Pollok\textsuperscript{1}  Dirk Zimmer\textsuperscript{1}  Francesco Casella\textsuperscript{2}

\textsuperscript{1}Institute of System Dynamics and Control, German Aerospace Center (DLR), Germany, \{alexander.pollok,dirk.zimmer\}@dlr.de
\textsuperscript{2}Dipartimento di Elettronica, Informazione e Bioingegneria, Politecnico di Milano, Italy, francesco.casella@polimi.it

Abstract

Most dynamic systems with a basis in nature can be described using Differential-Algebraic Equations (DAE), and hence be modelled using the modelling language Modelica. However, the concept of DAEs can still be generalised, when differential operators of non-integer order are considered. These so called fractional order systems have counterparts in naturally occurring systems, for instance in electrochemistry and viscoelasticity. This paper presents an implementation of approximate fractional-order differential operators in Modelica, increasing the scope of systems that can be described in a meaningful way. Properties of fractional-order systems are discussed and some approximation methods are presented. An implementation in Modelica is proposed for the first time. Several testing procedures and their results are displayed. The work is then illustrated by the application of the model to several physically motivated examples. A possible usability-enhancement using the concept of “Calling Blocks as functions” is suggested.

Keywords: Fractional Order Systems, fractional calculus, Integer-Order Approximations

1 Introduction

In Modelica, models are represented as Differential-Algebraic Equations, i.e., equations of the form $F(\dot{x}(t), x(t), t) = 0$. This formulation is adequate for most physical systems that can be described (or at least approximated) with a finite number of states. There are, however, some systems, where a more general but ultimately similar framework is needed: If fractional derivatives occur, the traditional DAE formulation is inadequate.

Fractional calculus, a misnomer\textsuperscript{1}, is a branch of mathematics that deals with non-integer powers of differentiation operators. The introduction to this concept is much simpler in the Laplace-domain. Normally, the Laplace variable is restricted to integer values. In fractional calculus, this restriction is lifted. Let us imagine the bode diagrams of the derivative operator $(s = s^1)$, the unity operator $(1 = s^0)$ and the half-derivator $(s^{0.5})$. The amplitude plot of the half-derivator has a slope of 10dB per decade, while the phase angle is constant at 45 degrees. This is illustrated in Figure 1. A detailed discussion of fractional calculus is given by Sabatier et al. (2007).

![Figure 1. Illustration of fractional derivatives](image)

\textsuperscript{1}this generalisation of differentiation operators is not restricted to fractions

Going back to time-domain, fractional differential operators can be defined in various ways, with the Caputo definition (Caputo, 1967) being applied in this paper. Contrary to the often preferred Riemann-Liouville definition, the Caputo definition allows for a physically meaningful initialisation of the operator. The Caputo Fractional Derivative of order $\alpha$ is defined as

$$\mathcal{L}^{-1}(s^\alpha) = D^\alpha f(t) := \frac{1}{\Gamma(m-\alpha)} \int_0^t \frac{f(\tau)}{(t-\tau)^{\alpha+1-m}} d\tau$$

with $f : \mathbb{R} \to \mathbb{R}$ being a continuous, differentiable function, the gamma function $\Gamma$, $\alpha \in \mathbb{R}, 0 < \alpha$ and $m \in \mathbb{Z}^+, m = \text{ceil}[\alpha]$.

Fractional-order systems occur naturally in various fields, like electrochemistry (Debnath, 2003), viscoelasticity (Koeller, 1984), heat diffusion (Povstenko, 2004) and biology (Magin, 2004). For example, exact solutions...
for local temperature and heat flux at the boundary of a semi-infinite body, using fractional calculus, are given by Kulish and Lage (2000) (for a Modelica Standard Library (MSL)-friendly implementation of this findings see Subsection 3.1). Another application is shaping noise frequency content according to given spectra (Klöckner et al., 2015). The usefulness is not limited to the modelling of PDE’s though, fractional order modelling can be used to describe the dynamics of scale-free networks, for instance (Goodwine and Leyden, 2015).

The goal of this paper is to show how fractional-order systems can be modelled using Modelica. The derivation, implementation and testing of suitable approximations in Modelica is illustrated in Section 2. Applications of this implementations are shown in Section 3. At last, the contents of the paper are discussed and possible ramifications to the Modelica languages are addressed in Section 4.

2 Fractional Order Modelling

2.1 Implementation

In Modelica, only the der()-operator has to be generalized to model arbitrary fractional order systems. Unfortunately, defining a new operator fractioner(state, order) is not possible based on the Modelica Language Specification 3.3. Therefore, the proposed implementation uses a Single-Input Single-Output block instead.

Fractional order systems have infinite dimensional transfer functions, and therefore have infinite memory (Vinagre et al., 2000). It is, however, possible to find reasonable approximations if the frequency range of interest is bounded (for a detailed error analysis refer to Pan and Das (2012)). Accordingly, as long as the modeller is not interested in extremely stiff systems, these approximations are adequate.

For implementations based on equation-based modelling languages, integer-order continuous approximations are the most useful. For examples, see Carlson and Halijak (1964), Xue et al. (2006) or Oustaloup et al. (2000). Some other methods are described in Vinagre et al. (2000), but not mentioned here, as their implementation in Modelica proved difficult due to a lack of user-level symbolic manipulation capabilities.

For Carlson’s method, Oustaloup’s method and Xue’s method, we found general symbolic expressions for the approximating transfer-functions, and implemented them in Modelica. Preliminary analysis showed that Oustaloup’s method was superior regarding flexibility and accuracy. For this reason, in the following only Oustaloup’s method and its implementation details are presented.

The integer-order approximation of a fractional operator by Oustaloup’s method is given in the Laplace domain by

\[
\lambda^\gamma G(s) = \omega_b^\gamma \cdot \prod_{k=-N}^{N} \frac{s + \omega_k}{s + \omega_k^\gamma} \quad (1)
\]

with the fitting range \((\omega_b, \omega_h)\), the fraction of differentiation \(\lambda\) and the order of approximation \(N\).

An important thing to notice is that \(\lambda\) is not bounded, so it is possible to simulate the second integral using \(\lambda = -2\), for example. However, more accurate results can be obtained if \(abs(\lambda)\) is kept low and surplus differential operations are simulated directly using the standard Modelica-notation.

We recreated the construction rule for the Oustaloup-Approximator in Modelica using linked first-order elements. The corresponding code can be seen in Listings 1, 2 and 3.

2.2 Testing

To test the validity and accuracy of the derived models, we applied three different testing scenarios: bode diagram, step response, and harmonic displacement. These tests are described in the following.

2.2.1 Bode Diagram

A Bode diagram of the unity operator (1 in the Laplace domain) is a straight line with amplitude 1 and phase angle zero degrees over the complete frequency range. The differentiator \((s)\) in the Laplace domain) has an positive slope of 20dB per decade and +90 degrees phase angle. Other operators like \(s^2\) or \(s^{-1}\) behave analogous. From this, we require the half-differentiator \(s^{0.5}\) to feature an ascending amplitude of 10dB per decade and +45 degrees phase angle.

In Figure 2 and Figure 3, bode plots of the implemented model with approximation orders 2 and 4 and fitting range \((0.001Hz,1000Hz)\) are presented.

It can be seen that slope of the amplitude shows a close fit to the required 10dB per decade. The phase angle shows pronounced ripple effects in the case of the 2nd order approximation, but no visible ripples in the case of the 4th order approximation.

The required amplitude values are matched in the complete fitting range. For the phase angle the acceptable range is somewhat smaller.

If the fitting interval is increased to cover a broader range of frequencies (not shown here), noticable ripples in the phase plot appear even for the 4th order approximation.
Listing 1. Excerpt of Modelica code for a fractional derivative operator

```modelica
block OustaloupOperator
  import Modelica.Blocks.Types.Init;
  parameter Integer order(min=1, max=4) = 4 "order of approximation (1,2,3,4)";
  parameter Real lambda = 0.5 "exponent of operator (-1=integrator, 1=derivative)";
  parameter Real w_lower(max=1) = 0.001 "lower fitting frequency [1/s]";
  parameter Real w_upper(min=1) = 1000 "higher fitting frequency [1/s]";
  parameter Modelica.Blocks.Types.Init initType="Init.InitialState" "Type of initialization (i: no init, 2: steady state, 3: initial state, 4: initial output)"
  annotation(Evaluate=true, Dialog(group=""Initialization""));
  parameter Real x_start[number]=zeros(number) "Initial or guess values of states"
  annotation(Dialog(group=""Initialization""));
  parameter Real y_start=0 "Initial value of output"
  annotation(Dialog(enable=initType == Init.InitialOutput, group=""Initialization""));
  final parameter Real wb = w_lower*Modelica.Const.pi;
  final parameter Real wh = w_upper*Modelica.Const.pi;
  final parameter Integer number = 1 + order*2;
  final parameter Real K = w^lambda;
  final parameter Real wk[number] = {FractionalOrder.Approximations.Internal.wk(i, wb, wh, order, lambda)
    for i in -order:order};
  final parameter Real wks[number] = {FractionalOrder.Approximations.Internal.wks(i, wb, wh, order, lambda)
    for i in -order:order};
  Real y_internal[number];
  Real x_internal[number];

  Modelica.Blocks.Interfaces.RealInput u
  annotation(Placement(transformation(extent={-120, -10}, {-100, 10})�);
  Modelica.Blocks.Interfaces.RealOutput y
  annotation(Placement(transformation(extent={100, -10}, {120, 10})�);

  equation
der(x_internal[1]) = -wk[1]*x_internal[1] + (wks[1]-wk[1]) * u*K;
y_internal[1] = x_internal[1] + y*K;

  for i in 2: number loop
der(x_internal[i]) = -wk[i]*x_internal[i] + (wks[i]-wk[i]) * y_internal[i-1];
y_internal[i] = x_internal[i] + y_internal[i-1];
end for;
y = y_internal[number];

  initial equation
if initType == Init.SteadyState then
der(x_internal) = zeros(number);
elseif initType == Init.InitialState then
  x_internal = x_start;
elseif initType == Init.InitialOutput then
  y = y_start;
end if;
end OustaloupOperator;
```

Listing 2. First internal function for the generation of coefficients

```modelica
function wks
    extends Modelica.Icons.Function;
    input Integer k;
    input Real wb;
    input Real wh;
    input Real N;
    input Real lambda;
    output Real wks;
algorithm
    wks := wb * (wh/wb)^((k+N+(1-lambda)/2)/(2*N+1));
end wks;
```

Listing 3. Second internal function for the generation of coefficients

```modelica
function wk
    extends Modelica.Icons.Function;
    input Integer k;
    input Real wb;
    input Real wh;
    input Real N;
    input Real lambda;
    output Real wk;
algorithm
    wk := wb * (wh/wb)^((k+N+(1+lambda)/2)/(2*N+1));
end wk;
```

2.2.2 Step Response

The step responses of the unity operator $y(t) = u(t)$ and the integrator $\dot{y}(t) = u(t)$ are known to be $y(t) = 1$ and $y(t) = t$ respectively, neglecting the initial conditions. Simultaneously, the unity operator and the integrator are identified in the Laplace Domain by $s^0$ and $s^{-1}$. The step responses of the fractional derivatives defined by $s^\lambda$ with $-1 \leq \lambda \leq 0$ have to constitute the continuous transition between those known step responses (Oldham, 1974).

The implemented model was instantiated 6 times and assigned $\lambda$-values in 0.2 intervals between -1 and 0. All models were subjected to a unit step (implemented by setting the input to 1 and the initial states of the models to 0). The results of this test can be seen in Figure 4.

It can be seen that the 6 step-responses form a smooth transition, and the outer ones correspond to the known...
step-responses mentioned earlier. All curves also match the curves given in Oldham (1974).

2.2.3 Harmonic Displacement

If a harmonic function is derived or integrated, the resulting function is a new harmonic function with a phase offset. For the second test, it is required that the result of fractionally integrated or derived harmonic functions behaves analogous.

The implemented model was instantiated 8 times and assigned $\lambda$-values in intervals of size 0.5 between -1 and 2.5. All models were subjected to a cosine input. Initialisation was done in such a way as to avoid unnecessary large initial transients: The models with derivative character would see the onset of the cosine input as a step and correspondingly respond with an impulse. For this reason, their initial states were set to a steady state solution. Models with integrative character were set to zero initial conditions, to set their integration constants to zero as well. The results of this test can be seen in Figure 5.

Figure 5. Cosine response of $s^{-1.0}$, $s^{-0.5}$, $s^{0.0}$, $s^{0.5}$ (zero state initialisation) and $s^{1.0}$, $s^{1.5}$, $s^{2.0}$, $s^{2.5}$ (steady state initialisation) approximated with the 3rd order Oustaloup’s Method

It can be seen that all model outputs form harmonic functions. Also, the offsets between the functions are uniform. The initialisations of $s^{0.5}$ and $s^{2.5}$ are obviously not optimal, but after a few seconds those deviations vanish.

3 Examples

3.1 Heat Conduction

In Kulish and Lage (2000), relationships between temperature and heat flow rate at arbitrary locations in semi-infinite domains are developed. The temperature at a given time at the boundary is in this way given by

$$T(t) = \alpha^{1/2} \cdot \frac{\delta^{-1/2}Q(t)}{\delta t^{1/2}} + T_0$$

with the thermal conductivity $k$, the thermal diffusivity $\alpha$, the Area $A$, the heat flow rate $Q$, and the starting temperature $T_0$.

As can be seen in Listing 4, the corresponding implementation in Modelica is straightforward and compact.

Listing 4. Modelica implementation of a semi-infinite thermal domain

```
Approximations.OustaloupOperator halfInt(order=3, lambda=-0.5);
equation halfInt.u = heatPort.Q_flow;
heatPort.T = (alpha^(1/2)/(k*A*s) + halfInt.y) + T_0;
```

In Figure 6, the result of a simulation can be seen, where a semi-infinite block was subjected to a periodic rectangular heat flow rate at the boundary. The temperature at the boundary exhibits strong memory-effects, as would be expected from such a system.

Figure 6. Temperature response of a semi-infinite domain subjected to periodic rectangular heat flow

3.2 Viscoelasticity

The dynamic behaviour of viscous fluids is commonly described with the Navier-Stokes equations. For the dynamic behavior of linear elastic materials, the Lame-Navier equations are used. Both equations have some similarities. As an example, let us take a look at the respective relationships between stress/velocity and
stress/strain for incompressible fluids and linear-elastic solids:

\[
\tau = \mu (\nabla v + \nabla v^T) \\
\sigma = \frac{1}{2} C (\nabla u + \nabla u^T)
\] (4)

with the stress tensors \( \tau \) and \( \sigma \), the viscosity \( \mu \), the tensor of elasticity \( C \), and the velocity and displacement tensors \( v \) and \( u \). The structure of both equations essentially differs only by one differential operation, as the velocity is the derivative of the displacement w.r.t. time.

Likewise, a relationship between the stress and strain in one-dimensional viscoelastic materials was found in Stiassnie (1979).

\[
\sigma = k \cdot \frac{\delta^\alpha \varepsilon}{\delta t^\alpha} \quad (0 \leq \alpha \leq 1)
\] (5)

with the stress \( \sigma \), the material properties \( k \) and \( \alpha \), and the strain \( \varepsilon \). For \( \alpha \)-values of 0 and 1, the behaviour of pure solids and fluids is obtained.

As with the other example, the implementation of this model in Modelica only takes two lines of code (not shown here).

The response of a viscoelastic block (\( \alpha = 0.45 \)) under constant tension can be seen in Figure 7. The result is similar to the results presented by Stiassnie (1979), where the model is validated against real-world measurements.

![Figure 7. tension-response of a viscoelastic block with a length of 1m under constant tension](image)

### 4 Discussion

The last sections showed that DAE systems containing time-derivatives of fractional order can be successfully implemented by the given means of the Modelica language. The provided solution offers an approximation that is good enough for at least a large set of technical applications. However, the given examples indicate that the typical application of fractional order derivatives is on the textual modelling level and that the applied formulation is, although practically feasible, still more clumpy than actually necessary. The reason for this is the implementation in block-form. This leads to a declaration that has a dummy character since it is only being used to textually connect its input and output. Because any implementation of a fractional-order operator requires an internal state, an implementation as function is not possible. Yet, it would be very helpful for the modeller if he could call the block like it would be a function. In concrete terms, this means that an anonymous declaration of a block in the equation section is enabled whose inputs and outputs are connected within the declaration statement. To illustrate this mechanism, let us revisit the example of Listing 4:

**Listing 5.** Modelica implementation of a semi-infinite thermal domain using the “calling blocks as function”-approach

```plaintext
block halfInt =
  Approximations.OustaloupOperator (order=3, lambda=0.5);

equation
  heatPort.T = (alpha^(1/2))/k
  + halfInt(u=heatPort.Q_flow).y/A + T_0;
```

Listing 5 presents a reformulation of Listing 3 based on the concept “Calling blocks as function”. First, a local declaration of a half-integrator is created from the general Operator-block. Then the expression `halfInt(u=heatPort.Q_flow).y` is represented as an anonymous declaration of the halfInt block. Within the parantheses, the input is connected and the .y states that the expression as a whole represents the output signal of the block. The presented concept “Calling Blocks as Function” is not a new idea. Different syntactical variants are currently in discussion within the Modelica Association based on contributions by Martin Otter, Hans Olsson, Peter Fritzson, Michael Sasena, Martin Sjölund and others. An implementation variant in an experimental equation-based language can be found for instance in Sol (Zimmer, 2010). Should this feature become part of a future Modelica language version, modelling with fractional order time-derivatives will be almost as convenient as with standard time derivatives.

### 5 Conclusion

By design, the Modelica language is limited to the use of integer-order differential operators. This excludes the modelling of certain physical systems. We present an implementation of Oustaloup’s approximation method in Modelica. The resulting model approximates fractional-order differential operators. Parameters for approximation order and frequency fitting range can be used to tailor the model to a specific application. In this way, the mentioned limitation of the Modelica language can be conveniently bypassed, thus increasing the scope of physical systems that can be described in a meaningful manner.
Reproducible research

The results of this paper can be reproduced using the code which is made available on:
github.com/DLR-SR/FractionalOrder

References


